Applications of DEs:
Simple LRC circuits

Prof. Joyner[1]

An LRC circuit is a closed loop containing an inductor of L henries, a resistor of R ohms, a capacitor of C farads, and an EMF (electro-motive force), or battery, of $E(t)$ volts, all connected in series.

They arise in several engineering applications. For example, AM/FM radios with analog tuners typically use an LRC circuit to tune a radio frequency. Most commonly a variable capacitor is attached to the tuning knob, which allows you to change the value of C in the circuit and tune to stations on different frequencies [R].

We use the following "dictionary" to translate between the diagram and the DEs.

EE object	term in DE (the voltage drop)	units	symbol
charge	$q = \int i(t)\,dt$	coulombs	
current	$i = q'$	amps	
emf	$e = e(t)$	volts V	
resistor	$Rq' = Ri$	ohms Ω	
capacitor	$C^{-1}q$	farads	
inductor	$Lq'' = Li'$	henries	

Kirchoff's First Law: The algebraic sum of the currents travelling into any node is zero.

Kirchoff's Second Law: The algebraic sum of the voltage drops around any closed loop is zero.

[1]These notes licensed under Attribution-ShareAlike Creative Commons license, `http://creativecommons.org/about/licenses/meet-the-licenses`. The diagrams were created using Dia `http://www.gnome.org/projects/dia/` and GIMP `http://www.gimp.org/` by the author. Originally written 9-21-2007. Last updated 2008-11-28.

Generally, the charge at time t on the capacitor, $q(t)$, satisfies the DE

$$Lq'' + Rq' + \frac{1}{C}q = e(t). \tag{1}$$

When there is no EMF, sometimes the following terminology is used. If $R > 2\sqrt{L/C}$ ("R is large") then the circuit is called **overdamped**. If $R = 2\sqrt{L/C}$ ("R is large") then the circuit is called **critally-damped**. If $0 \geq R < 2\sqrt{L/C}$ ("R is large") then the circuit is called **underdamped**.

Example 1: Consider the simple LC circuit given by the diagram in Figure 1.



Figure 1: A simple LC circuit.

This is a simple modelillustrating the idea of a radio tuner (the variable capacitor) which can tune into only two stations, "channel 2" and "chennel 11".

According to Kirchoff's 2^{nd} Law and the above "dictionary", this circuit corresponds to the DE

$$q'' + \frac{1}{C}q = \sin(2t) + \sin(11t).$$

The homogeneous part of the solution is

$$q_h(t) = c_1 \cos(t/\sqrt{C}) + c_1 \sin(t/\sqrt{C}).$$

If $C \neq 1/4$ and $C \neq 1/121$ then

$$q_p(t) = \frac{1}{C^{-1} - 4}\sin(2t) + \frac{1}{C^{-1} - 121}\sin(11t).$$

When $C = 1/4$ and the initial charge and current are both zero, the solution is

$$q(t) = -\frac{1}{117}\sin(11t) + \frac{161}{936}\sin(2t) - \frac{1}{4}t\cos(2t).$$

SAGE

```
sage: t = var("t")
sage: q = function("q",t)
sage: L,R,C = var("L,R,C")
sage: e = lambda t:sin(2*t)+sin(11*t)
sage: de = lambda y: L*diff(y,t,t) + R*diff(y,t) + (1/C)*y-e(t)
sage: L,R,C=1,0,1/4
sage: de(q(t))
diff(q(t), t, 2) - sin(11*t) - sin(2*t) + 4*q(t)
sage: desolve(de(q(t)),[q,t])
(-4*sin(11*t) - 117*t*cos(2*t))/468 + k1*sin(2*t) + k2*cos(2*t)
sage: desolve_laplace(de(q(t)),["t","q"],[0,0,0])
'-sin(11*t)/117+161*sin(2*t)/936-t*cos(2*t)/4'
sage: soln = lambda t: -sin(11*t)/117+161*sin(2*t)/936-t*cos(2*t)/4
sage: P = plot(soln,0,10)
sage: show(P)
```

This is displayed in Figure 2.



Figure 2: A LC circuit, with resonance.

You can see how the frequency $\omega = 2$ dominates the other terms.

When $0 < R < 2\sqrt{L/C}$ the homogeneous form of the charge in (1) has the form

$$q_h(t) = c_1 e^{\alpha t}\cos(\beta t) + c_2 e^{\alpha t}\sin(\beta t),$$

where $\alpha = -R/2L < 0$ and $\beta = \sqrt{4L/C - R^2}/(2L)$. This is sometimes called the **transient part** of the solution. The remaining terms in the charge are called the **steady state terms**.

Example: An LRC circuit has a 1 henry inductor, a 2 ohm resistor, 1/5 farad capacitor, and an EMF of $50\cos(t)$. If the initial charge and current is 0, since the charge at time t.

The IVP describing the charge $q(t)$ is

$$q'' + 2q' + 5q = 50\cos(t), \quad q(0) = q'(0) = 0.$$

The homogeneous part of the solution is

$$q_h(t) = c_1 e^{-t}\cos(2t) + c_2 e^{-t}\sin(2t).$$

The general form of the particular solution using the method of undetermined coefficients is

$$q_p(t) = A_1\cos(t) + A_2\sin(t).$$

Solving for A_1 and A_2 gives

$$q_p(t) = -10e^{-t}\cos(2t) - \frac{15}{2}e^{-t}\sin(2t).$$

SAGE

```
sage: t = var("t")
sage: q = function("q",t)
sage: L,R,C = var("L,R,C")
sage: E = lambda t: 50*cos(t)
sage: de = lambda y: L*diff(y,t,t) + R*diff(y,t) + (1/C)*y-E(t)
sage: L,R,C = 1,2,1/5
sage: de(q(t))
diff(q(t), t, 2) + 2*diff(q(t), t, 1) + 5*q(t) - 50*cos(t)
sage: desolve_laplace(de(q(t)),["t","q"],[0,0,0])
```

```
'%e^-t*(-15*sin(2*t)/2-10*cos(2*t))+5*sin(t)+10*cos(t)'
sage: soln = lambda t:\
 e^(-t)*(-15*sin(2*t)/2-10*cos(2*t))+5*sin(t)+10*cos(t)
sage: P = plot(soln,0,10)
sage: show(P)
sage: P = plot(soln,0,20)
sage: show(P)
sage: soln_ss = lambda t: 5*sin(t)+10*cos(t)
sage: P_ss = plot(soln_ss,0,10)
sage: soln_tr = lambda t: e^(-t)*(-15*sin(2*t)/2-10*cos(2*t))
sage: P_tr = plot(soln_tr,0,10,linestyle="--")
sage: show(P+P_tr)
```

This plot (the solution superimposed with the transient part of the solution) is displayed in Figure 3.



Figure 3: A LRC circuit, with damping, and the transient part (dashed) of the solution.

Exercise: Use SAGE to solve

$$q'' + \frac{1}{C}q = \sin(2t) + \sin(11t), \quad q(0) = q'(0) = 0,$$

in the case $C = 1/121$.

References

[KL] Wikipedia entry for Kirchhoff's laws: http://en.wikipedia.org/wiki/Kirchhoffs_circuit_laws

[K] Wikipedia entry for Kirchhoff: http://en.wikipedia.org/wiki/Gustav_Kirchhoff

[N] Wikipedia entry for Electrical Networks: http://en.wikipedia.org/wiki/Electrical_network

[R] General wikipedia introduction to LRC circuits: http://en.wikipedia.org/wiki/RLC_circuit